
Résumé

Ce mémoire traite d’une gestion de production et d’une minimisation de coûts de
production d’une entreprise produisant des desserts. L’objectif est de déterminer le
coût de production minimum en optimisant l’achat des matières premières, le plan
de production, le plan de stockage, les engagements et les licenciements au début
de chaque saison afin de satisfaire la demande tout en respectant les différentes
contraintes. Ensuite, la réponse a deux questions d’analyse post-optimale est donnée
où l’on demande le prix maximum que l’on est prêt à payer pour augmenter la
capacité de stockage durant une saison d’une part et pour l’acquisition d’unités de
matière première supplémentaires d’autre part.



Remerciements

Ce mémoire a été réalisé sur une année et a été soutenu par certaines personnes
que je souhaite remercier.

Premièrement, je tiens à remercier le Professeur Daniel De Wolf qui a été à ma
disposition tout au long de l’année en donnant son temps et son écoute. Et qui m’a
guidé et conseillé quand j’en avais besoin afin d’avancer dans ce travail.

Je souhaite également remercier toutes les personnes qui m’ont aidé de près ou de
loin dans l’accomplissement de ce mémoire.



Table des matières

Liste des figures II

Liste des tableaux III

1 Introduction 1

2 Données du problème 3
2.1 Indices . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 3
2.2 Données du problème . . . . . . . . . . . . . . . . . . . . . . . . . . 4

2.2.1 Données associées à la demande . . . . . . . . . . . . . . . . 4
2.2.2 Données associées aux matières premières . . . . . . . . . . 4
2.2.3 Données associées au stockage . . . . . . . . . . . . . . . . . 5
2.2.4 Données associées à la fabrication . . . . . . . . . . . . . . . 6
2.2.5 Données associées au personnel . . . . . . . . . . . . . . . . 7

3 Formulation du problème 9
3.1 Variables . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 9
3.2 Expression de l’objectif . . . . . . . . . . . . . . . . . . . . . . . . . 10
3.3 Contraintes . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 11

3.3.1 Contraintes associés à la demande . . . . . . . . . . . . . . . 11
3.3.2 Contraintes associées au stock . . . . . . . . . . . . . . . . . 11
3.3.3 Contraintes associées à la production . . . . . . . . . . . . . 12
3.3.4 Contraintes associées au personnel . . . . . . . . . . . . . . 12
3.3.5 Positivité des variables . . . . . . . . . . . . . . . . . . . . . 12

iv



TABLE DES MATIÈRES I

4 Mise en oeuvre avec GAMS 13
4.1 Résultats . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 13

4.1.1 Coût total minimum obtenu . . . . . . . . . . . . . . . . . . 14
4.1.2 Résultats optimaux . . . . . . . . . . . . . . . . . . . . . . . 14

4.2 Analyse Post-optimale . . . . . . . . . . . . . . . . . . . . . . . . . 17

5 Conclusion 19

Bibliographie 21

Annexes 21

A Code GAMS 22

B Listing de résultats GAMS 27



Table des figures

4.1 Résultats optimaux pour l’achat des matières premières par saison . 17
4.2 Résultats optimaux pour le stockage par saison . . . . . . . . . . . 18

II



Liste des tableaux

2.1 Indices . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 3
2.2 Demande des deux desserts par saison en unités . . . . . . . . . . . 4
2.3 Offre disponible des matières premières par saison en kg. . . . . . . 4
2.4 Quantité d’ingrédients nécessaires par dessert en kg. . . . . . . . . . 4
2.5 Prix de la matière première par saison en €/kg . . . . . . . . . . . . 5
2.6 Coût de stockage des éléments pendant une saison en € . . . . . . . 5
2.7 Volume de stockage de chaque élément en litres . . . . . . . . . . . 6
2.8 Stocks initiaux . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 6
2.9 Temps de confection et de conditionnement de chaque dessert . . . 6
2.10 Coût Horaire par atelier en € par heure . . . . . . . . . . . . . . . . 7
2.11 Nombre de jours ouvrables par saison . . . . . . . . . . . . . . . . . 7
2.12 Nombre initial d’ouvriers dans chaque atelier . . . . . . . . . . . . . 7
2.13 Données concernant les coûts d’engagement et de licenciement en

fonction de l’atelier en € . . . . . . . . . . . . . . . . . . . . . . . . 7
2.14 Nombre maximum d’ouvriers dans les ateliers . . . . . . . . . . . . 8

4.1 Tableau des coûts . . . . . . . . . . . . . . . . . . . . . . . . . . . . 14
4.2 Tableau des résultats : Offre et demande . . . . . . . . . . . . . . . 15
4.3 Tableau des résultats : Stockage . . . . . . . . . . . . . . . . . . . . 15
4.4 Tableau des résultats : Main d’oeuvre . . . . . . . . . . . . . . . . . 16

III



Chapitre 1

Introduction

Ce travail de fin d’études portera sur la modélisation et l’optimisation d’un pro-
blème de production.

Une entreprise est spécialisée dans la production de desserts glacés, elle produit deux
desserts : le "Délices du Cacao" et le "Délices du Verger" fabriqués respectivement
à base de chocolat et de fruits frais, les matières premières. La particularité de ces
produits est qu’ils suivent une tendance saisonnière, le "Délice du Cacao" étant
consommé principalement en été et le "Délices du Verger" étant consommé princi-
palement en hiver. Il faut déterminer la quantité de fruits frais et de chocolat à se
procurer en fonction de différents paramètres : la demande, la capacité de stockage,
les capacités de production, les productions effectuées aux périodes antérieures. Le
prix du chocolat ne varie pas le long de l’année et l’offre est virtuellement illimitée.
Par contre, le prix des fruits frais varie en fonction de la saison, il est plus élevé en
hiver lorsqu’il n’y a que peu d’offre. Il est le plus faible en été lorsque la production
de fruits est forte. Il faudra tenir compte de cette subtilité car le moment où la
demande de "Délices du Verger" est la plus élevée n’est pas le moment où l’offre de
fruits frais est la plus élevée. Il faudra anticiper et faire des stocks au bon moment
afin de réduire les coûts de stockage et d’achat.

1



CHAPITRE 1. INTRODUCTION 2

Une fois achetés, les fruits frais se conservent dans un surgélateur de capacité limitée,
le chocolat lui, est stocké dans un entrepôt. Les desserts finaux sont également
stockés dans ce surgélateur et chaque élément stocké a un coût. Il faut noter qu’il
y a un stock initial de desserts ainsi que de fruits frais présents dans le surgélateur
en début d’année. Il y a également un stock initial de chocolat.

Chaque élément de matière première passe par deux ateliers différents avant d’être
transformés en desserts. Premièrement, la matière première passe dans l’atelier de
confection, ensuite dans un atelier de conditionnement. Le temps passé dans chaque
atelier pour chaque produit est différent, typiquement, la confection est plus lente
que le conditionnement. Chaque atelier demande une qualification du personnel
différente et un coût horaire différent. L’atelier de fabrication demande une qua-
lification plus élevée et à un coût horaire plus élevé que l’atelier de conditionnement.

Les ateliers de confection et de conditionnement doivent donc travailler en fonction
de la demande prévue de chaque dessert. Si la capacité d’un atelier n’est pas
suffisante pour subvenir à la demande, il faut produire cette quantité à l’avance
et stocker les desserts produits dans le surgélateur. Il est possible d’engager des
ouvriers si la capacité de production n’est pas suffisante, mais il y a un nombre
maximum d’ouvriers à respecter dans chaque atelier. A l’opposé, il est possible de
licencier des ouvriers si ils sont trop nombreux pour la production demandée, mais
cela a un coût qui dépend du niveau de qualification que chaque ouvrier.

Le but est de déterminer le plan de production optimal sur une année pour cette
entreprise. Il faut déterminer les achats de matières premières, le plan de production,
le plan de stockage, les engagements et licenciements en début de chaque trimestre
afin de satisfaire la demande tout en gardant le coût de production au minimum.
Le coût de production est composé du coût d’achat des matières premières, du
coût de stockage, du coût de la main d’oeuvre et du coût des engagements et des
licenciements.



Chapitre 2

Données du problème

Dans cette section, les données initiales du problème sont expliquées. D’abord, des
indices sont précisés et seront utiles lors de la détermination des variables. L’énoncé
du mémoire est source de ces données (De Wolf, 2019a).

2.1 Indices

Les indices utilisés dans la formulation de l’objectif sont définis dans le tableau 2.1

Définition Indice 1 2 3 4
Matière première mp Chocolat Fruits frais

Dessert d "Délices du Cacao" "Délices du Vergers"
Saisons t Printemps Eté Automne Hiver

Atelier de confection a Confection Conditionnement

Table 2.1 – Indices

3

2.2 Données du problème

2.2.1 Données associées à la demande

Le tableau 2.2 donne la Demande (DEMd,t) des deux desserts par saison en unités.

Demande Printemps Eté Automne Hiver
"Délice du Cacao" (unités) 4500 7500 7000 6000
"Délice du Verger" (unités) 2500 1500 5000 7000

Table 2.2 – Demande des deux desserts par saison en unités

2.2.2 Données associées aux matières premières

Le tableau 2.3 donne l’Offre (OFFmp,t) maximum des matières premières (en kg)
disponibles par saison. L’offre de chocolat est virtuellement infinie.

Offre Printemps Eté Automne Hiver
Chocolat (kg) illimitée illimitée illimitée illimitée
Fruits frais (kg) 1000 2000 500 60

Table 2.3 – Offre disponible des matières premières par saison en kg.

Le tableau 2.4 donne la Quantité d’ingrédients (QId,mp) utilisées pour chacun des
deux desserts. Le dessert "Délices du Verger" n’utilise que des fruits frais et le
dessert "Délices du Cacao" n’utilise que du chocolat.

Dessert Cacao (kg) Fruits (kg)
"Délice du Cacao" 0.05 0
"Délice du Verger" 0 0.1

Table 2.4 – Quantité d’ingrédients nécessaires par dessert en kg.

Le tableau 2.5 donne le Prix (Pmp,t) de chaque ingrédient en fonction de la saison.
L’offre plus faible de fruits frais en hiver se traduit pas un prix plus élevé et
vice-versa.

Prix Printemps Eté Automne Hiver
Prix du chocolat (€/kg) 6.5 6.5 6.5 6.5

Prix des fruits frais (€/kg) 1.5 1 2 4

Table 2.5 – Prix de la matière première par saison en €/kg

2.2.3 Données associées au stockage

La capacité (CAPA) du surgélateur où l’on stocke les fruits frais et les deux types
de desserts est de 2000 l.

Le tableau 2.6 donne respectivement le Coût de Stockage des Matières Premières
(CSMPmp) et le Coût de Stockage des Desserts (CSDd) pendant une saison.

Prix €
Du kg de chocolat 0.1
Du kg de fruits 0.35

(a) Matières premières

Prix €
Du "Délice du Cacao" 0.25
Du "Délice du Verger" 0.25

(b) Desserts

Table 2.6 – Coût de stockage des éléments pendant une saison en €

Le tableau 2.7 donne respectivement le volume occupé dans le surgélateur par
les Matières Premières (V MPmp) et par les Desserts (V Dd). Il faut noter que le
volume du chocolat est 0 car il n’est pas stocké dans le surgélateur.

Volume Litres
Du chocolat 0

Du kg de fruits frais 2.5

(a) Matières premières

Volume Litres
D’un "Délice du Cacao" 0.5
D’un "Délice du Verger" 0.6

(b) Desserts

Table 2.7 – Volume de stockage de chaque élément en litres

Le tableau 2.8 donne les Stocks Initiaux de Matière Première en kg (SIMPmp)
ainsi que les Stocks Initiaux de Desserts en unités (SIDd).

Stock initial Kg
Du chocolat 120
Du fruits frais 50

(a) Matières premières

Stock initial Unités
Du "Délice du Cacao" 250
Du "Délice du Verger" 350

(b) Desserts

Table 2.8 – Stocks initiaux

2.2.4 Données associées à la fabrication

Le tableau 2.9 donne les Temps de Fabrication (TFd,a)pour chaque dessert dans
chaque atelier en minutes par unité.

Temps en minutes pour la confection (en min) le conditionnement (en min)
Du "Délice du Cacao" 30 10
Du "Délice du Verger" 35 15

Table 2.9 – Temps de confection et de conditionnement de chaque dessert

Le tableau 2.10 donne les Coûts Horaires (CHa) dans chaque atelier. Il faut noter
que l’on considère 8 heures dans une journée.

Coût horaire €/heure
Dans l’atelier de confection 12,50

Dans l’atelier de conditionnement 9,50

Table 2.10 – Coût Horaire par atelier en € par heure

2.2.5 Données associées au personnel

Le tableau 2.11 donne le nombre de Jours ouvrables par saison (Jt).

Printemps Eté Automne Hiver
Jours ouvrables 60 63 57 60

Table 2.11 – Nombre de jours ouvrables par saison

Le tableau 2.12 donne le nombre d’ouvriers initial dans chaque atelier (PERSINITa).

Atelier confection Conditionnement
Personnel initial 30 12

Table 2.12 – Nombre initial d’ouvriers dans chaque atelier

Le tableau 2.13 donne les Coûts relatifs à l’Engagement (CENGa) ou le Licencie-
ment (CLICa) d’ouvriers par atelier. Le prix dépend du niveau de compétences
nécessaires de l’atelier.

Atelier Confection Conditionnement
Coût d’engagement (€) 1 000 500
Coût de licenciement (€) 2 500 1 500

Table 2.13 – Données concernant les coûts d’engagement et de licenciement en
fonction de l’atelier en €

Le tableau 2.14 donne le nombre maximal d’ouvriers (PERSMAXa) pouvant être
présent dans chaque atelier à tout moment.

Maximum d’ouvriers
Dans l’atelier de confection 52

Dans l’atelier de conditionnement 22

Table 2.14 – Nombre maximum d’ouvriers dans les ateliers



Chapitre 3

Formulation du problème

3.1 Variables

Voici le choix des variables afin de résoudre la problématique.

— QMPmp,t = la Quantité de Matière Première achetée par saison (kg/saison).

— QDd,t = la Quantité de Desserts produite par saison (unités/saison).

— SMPmp,t = Stock de Matière Première à la fin de la saison (kg).

— SDd,t = Stock de Dessert à la fin de la saison (unités).

— NOUV Ra,t = Nombre d’Ouvriers présents dans chaque atelier par saison.
Nous faisons l’hypothèse que des ouvriers peuvent être engagés à temps
partiel.

— EOUV Ra,t = Nombre d’Embauches d’ouvriers dans chaque atelier par saison.

— LOUV Ra,t = Nombre de Licenciements d’ouvriers dans chaque atelier par
saison.

— CAMP = Coût d’Achat des Matières Premières

— CSTOCK = Coût de Stockage

— CMO = Coût de la Main d’Oeuvre

— CLE = Coûts de Licenciements et d’Engagements

— CT = Coût Total

9

3.2 Expression de l’objectif

L’objectif étant de minimiser le coût de production de cette entreprise. Le coût est
divisé en 4 parties : le coût d’achat des matières premières, le coût de stockage,
le coût de la main d’oeuvre et le coût des licenciements et des engagements. Ces
coûts s’additionnent pour donner le coût total qui sera minimisé.

— Coût Achat Matières Premières. Il est calculé sur base de la quantité de
matière première achetée et de son coût unitaire.

CAMP =
2∑

mp=1

4∑
t=1

Pmp,t ∗QMPmp,t (3.1)

— Coût de Stockage. Il est calculé sur base de la quantité de matière première
stockée et du prix de stockage par unité de matière première. La logique est
identique pour les desserts.

CSTOCK =
2∑

mp=1

4∑
t=1

CSMPmp ∗ SMPmp,t +
2∑

d=1

4∑
t=1

CSDd ∗ SDd,t (3.2)

— Coût de la main d’oeuvre. Il est calculé sur base du coût horaire et du nombre
d’ouvriers présents. Ceci est multiplié par 8h/jour et par le nombre de jours
afin d’avoir le coût par mois.

CMO =
2∑

a=1

2∑
d=1

4∑
t=1

CHa ∗NOUV Ra,t ∗ 8 ∗ Jt (3.3)

— Coûts de licenciements et d’engagements. Il est calculé sur base du nombre
d’engagements/licenciements et du prix unitaire d’engagement/de licencie-
ments.

CLE =
2∑

a=1

4∑
t=1

CENGa ∗ EOUV Ra,t + CLICa ∗ LOUV Ra,t (3.4)

L’objectif principal étant de minimiser le coût total et s’exprime comme suit :

minCT = CAMP + CSTOCK + CMO + CLE (3.5)

3.3 Contraintes

La problème possède des contraintes qui sont exprimées ici.

3.3.1 Contraintes associés à la demande

La quantité de matière première achetée ne peut pas dépasser l’offre.

QMPmp,t ≤ OFFmp,t ∀mp, t (3.6)

3.3.2 Contraintes associées au stock

Initialisation des stocks de desserts et de matière première pendant l’hiver 0

SDd,0 = SIDd ∀d (3.7)

SMPmp,0 = SIMPmp ∀mp (3.8)

Les stocks de matière première en fin de saison sont la somme du stock restant
de la saison précédente et de la quantité de matière première achetée pendant la
saison. A laquelle on retire la quantité de la matière première utilisée pour faire un
dessert.

SMPmp,t = SPMmp,t−1 + QMPmp,t −
2∑

d=1
QTEmp,d ∗QDd,t ∀t, mp (3.9)

Le stocks de desserts en fin de saison sont la somme de la quantité de dessert
produite et le stock de dessert restant de la saison précédente. A laquelle on retire
la demande de dessert, donc les desserts que l’on a vendus.

SDd,t = QDd,t + SDd,t−1 −DEMd,t ∀d, t (3.10)

La capacité de stockage dans le surgélateur doit être plus élevée que le stockage
effectif en fin de saison. Ce stockage effectif est calculé avec le stock de dessert
restant et le stock de matière première restant, les deux en fonction du volume

qu’ils occupent dans le surgélateur.
2∑

d=1
V Dd ∗ SDd,t +

2∑
mp=1

V MPmp ∗ SMPmp,t ≤ CAPA ∀t (3.11)

3.3.3 Contraintes associées à la production

Il faut s’assurer que le nombre d’heures disponibles n’est pas dépassé. Le nombre
d’heures utilisées pour fabriquer la quantité de desserts doit être inférieure au
produit des jours disponibles en une saison, du nombre d’ouvriers présents dans les
ateliers et du nombre d’heures disponibles dans une journée, qui est au nombre de
8. Le temps de fabrication est divisé par 60 afin de le convertir en heures.

2∑
d=1

TFd,a

60 ∗QDd,t = Jt ∗ 8 ∗NOa,t ∀a, t (3.12)

3.3.4 Contraintes associées au personnel

Il y a une contrainte d’initialisation pour les ouvriers

NOUV Ra,0 = PERSINITa ∀a (3.13)

Il y a une limite au nombre d’ouvriers pouvant être présents dans les ateliers

NOUV Ra,t ≤ PERSMAXa ∀a, t (3.14)

Le nombre d’ouvriers par atelier est déterminé par le nombre d’ouvriers de la
saison précédente additionné de la différence entre les engagements et les licencie-
ments.

NOUV Ra,t = NOUV Ra,t−1 + EOUV Ra,t − LOUV Ra,t ∀a, t (3.15)

3.3.5 Positivité des variables

Toutes les variables mentionnées dans la section 3.1 sont supposées être positives
ou nulles. On suppose également que le nombre d’ouvriers ne doit pas être entier
car on accepte les emplois à temps partiel.



Chapitre 4

Mise en oeuvre avec GAMS

4.1 Résultats

Le problème présenté comme tel est un problème linéaire car toutes les variables
sont à la puissance 1 et ne se multiplient jamais entre elles. L’optimisation li-
néaire consiste à trouver la solution qui minimise l’objectif tout en respectant les
contraintes. Elle est fortement utilisée lors de gestion de production de produits.
(Arsham, 2015)

Un optimiseur linéaire (LP) utilisant la méthode du simplexe (CPLEX) a donc été
utilisé lors de la mise en oeuvre avec GAMS afin de trouver le coût minimum de
cette production de desserts tenant compte des données et des contraintes citées
plus tôt. L’algorithme du simplexe va

Le code GAMS a été réalisé avec l’aide de deux documents : Introduction au
langage GAMS (De Wolf, 2019b) et la documentation GAMS en ligne (GAMS
Documentation Center : Version 31 , s. d.). Le code en GAMS peut être trouvé
dans l’annexe A et le listing des résultats donné par GAMS se trouve dans l’annexe
B.

13

4.1.1 Coût total minimum obtenu

Les coûts totaux sont exprimés dans le tableau 4.1 contenant le coût d’achat
des matières premières (CAMP), le coût de stockage (CSTOCK), le coût de la
main d’oeuvre (CMO) et le coût des licenciements et des engagements (CLE). Ces
coûts s’additionnent pour donner le coût total (CT). Ces coûts sont minimums et
optimaux.

Il est intéressant de mentionner que le coût de la main d’oeuvre est le plus important,
elle représente 80.6 % du coût total.

Coût €
CAMP 9 550.65

CSTOCK 1 296.21
CMO 345 158.33
CLE 72 094.11
CT 428 099.31

Table 4.1 – Tableau des coûts

4.1.2 Résultats optimaux

Le tableau 4.2 suivant exprime le plan optimal d’achat des matières premières
(QMP) et le plan de production des desserts (QD) comparé à la demande de ceux ci.
Le plan optimal de stockage des matières premières (SMP) et des desserts (SD) est
décrit dans le tableau 4.3. Le nombre d’ouvriers (NOUVR), d’embauches (EOUVR)
et de licenciements (LOUVR) par atelier par saison sont décrits dans le tableau
4.4.

On peut noter que la production de "Délices du Cacao" est anticipée pendant le
printemps, mais elle correspond exactement à la demande pendant l’automne et
l’hiver. Ce n’est pas le cas pour les "Délices du Verger". Etant donné que la demande
est le plus forte en hiver, mais le prix des fruits frais est également plus haut, la
production de ces desserts s’est faite plus tôt dans l’année lorsque le prix des fruits
était bien plus bas et a ensuite été stockée.

Le chocolat n’est jamais stocké et ce en raison de sa disponibilité illimitée et de
son prix constant au long de l’année, il n’y a aucun avantage en termes de coût
d’acheter du chocolat en avance et de le stocker à 0.10€/kg. La durée de l’exercice
étant de 1 an, tous les stocks sont donc écoulés en hiver et le surgélateur est vide.
Le stockage dans le surgélateur est maximal pendant l’été car le prix des fruits
frais est le plus bas et l’offre est la plus haute durant cette saison, menant donc à
un stockage de fruits frais et de "Délices du Verger".

Il y également un grand nombre de licenciements dès le printemps, 70.6% des
ouvriers en confection et 73.25% des ouvriers en conditionnement sont licenciés. Si
l’on avait gardé le nombre d’ouvriers initiaux, la capacité de production serait trop
élevée.

Variable Printemps Été Automne Hiver
Matière première achetée (kg)

QMP (Chocolat) 166.58 300.91 350 300
QMP (Fruits frais) 183.80 771.20 500 60

Desserts produits (unités)
QD (Délices du Cacao) 5 731.71 6 018.30 7 000 6 000
QD (Délices du Verger) 2 337.98 2 454.88 4 717.95 6 139.20

Table 4.2 – Tableau des résultats : Offre et demande

Variable Printemps Été Automne Hiver
Stockage

SMP (Chocolat en kg) 0 0 0 0
SMP (Fruits frais en kg) 0 525.71 553.92 0

SD (Délices du Cacao en unités) 1 481.71 0 0 0
SD (Délices du Verger en unités) 187.98 1 142.86 860.81 0
Stockage total par saison (litres) 853.64 2 000 1 901.28 0

Table 4.3 – Tableau des résultats : Stockage

Variable Printemps Été Automne Hiver
Main d’oeuvre

NOUVR (Confection) 8.81 8.81 13.71 13.81
NOUVR (Conditionnement) 3.21 3.21 5.14 5.28

EOUVR (Confection) 0 0 4.90 0
EOUVR (Conditionnement) 0 0 1.94 0.14

LOUVR (Confection) 21.19 0 0 0
LOUVR (Conditionnement) 8.79 0 0 0

Table 4.4 – Tableau des résultats : Main d’oeuvre

4.2 Analyse Post-optimale

L’analyse post-optimale consiste à répondre à deux questions :

1. Jusqu’à quel prix seriez-vous prêts à payer des unités au delà des
500 kilos que peut vous fournir le fournisseur habituel en automne ?

Selon le rapport donné par GAMS montré dans la figure 4.1, le prix que l’on
est prêt à payer se trouve dans la colonne de valeur marginale qui représente le
prix caché. Pour chaque kilo de fruits frais acheté en plus en automne, le coût
total diminue de 19.83€. Il faut que le prix d’achat de cette unité soit inférieur
ou égal à 19.83€ afin que cela reste rentable. Il faut noter cependant que cette
valeur peut changer si le membre de droite de l’équation 3.6 augmente de
trop, il y a donc un nombre maximal d’unités en plus que l’on peut acheter
sans que ce prix caché ne change. La limite de cette augmentation n’est pas
disponible avec GAMS.

Figure 4.1 – Résultats optimaux pour l’achat des matières premières par saison

2. Quel prix êtes vous prêts à payer en été pour louer de la capa-
cité supplémentaire en surgélation auprès d’une société spécialisée
dans le stockage de produits réfrigérés ?

Selon le rapport donné par GAMS montré dans la figure 4.2, le prix que l’on
est prêt à payer se trouve dans la colonne de valeur marginale qui représente
le prix caché. Pour chaque litre de capacité supplémentaire, nous sommes

disposés à payer 8,19 € de plus. Il est à noter que l’été est la seule saison
où l’on a atteint la capacité maximal de stockage. La même remarque que
sur la question précédente est valable, cette valeur de 8,19€ peut changer si
l’on augmente le membre de droite de l’équation 3.11. Il y a une quantité
maximale de capacité de stockage que l’on peut louer en plus sans que cela
ne change le prix caché.

Figure 4.2 – Résultats optimaux pour le stockage par saison



Chapitre 5

Conclusion

Ce mémoire avait comme but de calculer et d’optimiser le plan de production de la
division "desserts glacés" d’une entreprise tout en minimisant les coûts totaux sur
l’exercice d’un an. Ce coût était dépendant de plusieurs paramètres. D’abord la
tendance saisonnière de la demande des deux desserts ainsi que l’offre variable des
matières premières. Ensuite de la capacité de stockage et du prix. Finalement de la
capacité de production dans les différents ateliers.

Le coût total minimum obtenu est de 428 099.31€ pour une année de production et
peux se diviser en plusieurs domaines : Le coût d’achat des matières premières est
de 9 550.65€, le coût de stockage est de 1 296.21€, le coût de la main d’oeuvre est
de 345 158.33€ et le coût des licenciements et des embauches est de 72 094.11€. On
peut remarquer que le coût de la main d’oeuvre est le plus important et représente
80.6% du coût total.

Vu la demande et l’offre fluctuante, la production a été anticipée principalement
au printemps pour le "Délices du Cacao" pour subvenir aux besoins de l’été. Et
elle a également été anticipée au niveau du "Délices du Verger" au printemps et
en été afin de subvenir aux besoins de l’automne et de l’hiver. Ceci mène donc à
un stockage principalement pendant le printemps pour le "Délice du Cacao" et au

19

printemps, été et automne pour le "Délice du Verger".

Le stockage étant maximal durant l’été, la question était de déterminer le prix
maximal que l’on était prêt à payer pour de la capacité de stockage supplémentaire
dans une entreprise spécialisée. Nous sommes prêts à payer 8.19€ de plus pour un
litre de capacité de stockage supplémentaire.

L’offre de fruits frais étant utilisée au maximum durant l’automne et l’hiver, la
question était de déterminer le prix que l’on était prêt à payer pour acheter des
unités au delà des 500kg que l’on peut obtenir en automne. Nous sommes prêts à
payer un maximum de 19.83€ pour que cela reste rentable.

Pour les deux questions d’analyse post-optimale, il faut noter que la valeur obtenue
peut changer si la valeur du membre de droite des équations 3.11 et 3.6 augmente de
trop. La valeur maximale exacte du membre de droite ne peut pas être déterminée
avec le logiciel GAMS.

Annexe A

Code GAMS

OPTION OPTCR = 0.0;

SETS

MP les matieres premieres /CHOCOLAT, FRUITS_FRAIS/

D les desserts /DELICES_DU_CACAO, DELICES_DU_VERGER/

T les saisons /HIVER0, PRINTEMPS, ETE, AUTOMNE, HIVER/

T1(T) les saisons /PRINTEMPS, ETE, AUTOMNE, HIVER/

A les ateliers de fabrication /CONFECTION, CONDITIONNEMENT/;

PARAMETERS

TABLE DEM(D,T) demande de dessert par saison (unites)

PRINTEMPS ETE AUTOMNE HIVER

DELICES_DU_CACAO 4500 7500 7000 6000

DELICES_DU_VERGER 2500 1500 5000 7000 ;

TABLE OFF(MP,T) Offre de matiere premiere par saison (kg)

PRINTEMPS ETE AUTOMNE HIVER

CHOCOLAT INF INF INF INF

FRUITS_FRAIS 1000 2000 500 60 ;

TABLE QI(D,MP) Quantite d ingredient necessaire par dessert (kg)

CHOCOLAT FRUITS_FRAIS

22

DELICES_DU_CACAO 0.05 0

DELICES_DU_VERGER 0 0.1 ;

TABLE P(MP,T) Prix de la matiere premiere par saison (euros par kg)

PRINTEMPS ETE AUTOMNE HIVER

CHOCOLAT 6.5 6.5 6.5 6.5

FRUITS_FRAIS 1.5 1 2 4 ;

TABLE TF(D,A) Temps de fabrication par dessert pour chaque atelier (minutes par unite)

CONFECTION CONDITIONNEMENT

DELICES_DU_CACAO 30 10

DELICES_DU_VERGER 35 15 ;

PARAMETERS

CSMP(MP) Cout de stockage par kg de matiere premiere (euros par kg par saison)

/CHOCOLAT 0.1

FRUITS_FRAIS 0.35/

CSD(D) Cout de stockage par dessert (euros par unite par saison)

/DELICES_DU_CACAO 0.25

DELICES_DU_VERGER 0.25/

VMP(MP) Volume d une unite de matiere premiere (litres)

/CHOCOLAT 0

FRUITS_FRAIS 2.5/

VD(D) Volume d une unite de dessert (litres)

/DELICES_DU_CACAO 0.5

DELICES_DU_VERGER 0.6/

SIMP(MP) Stock initial de matiere premiere (kg)

/CHOCOLAT 120

FRUITS_FRAIS 50 /

SID(D) Stock initial de dessert (unites)

/DELICES_DU_CACAO 250

DELICES_DU_VERGER 350/

CH(A) Cout horaire par atelier (euros par heure)

/CONFECTION 12.5

CONDITIONNEMENT 9.5/

J(T) Jours ouvrables par saison (jours)

/PRINTEMPS 60

ETE 63

AUTOMNE 57

HIVER 60/

PERSINIT(A) Nombre initial d ouvriers par atelier (personnes)

/CONFECTION 30

CONDITIONNEMENT 12/

CENG(A) Cout d engagement par atelier (euros par personne)

/CONFECTION 1000

CONDITIONNEMENT 500 /

CLIC(A) Cout de licenciement par atelier (euros par personne)

/CONFECTION 2500

CONDITIONNEMENT 1500/

PERSMAX(A) Nombre maximum d ouvriers par atelier (personnes)

/CONFECTION 52

CONDITIONNEMENT 22/ ;

SCALAR CAPA Capacité du surgélateur (litres) /2000/ ;

VARIABLES

QMP(MP,T) Quantite de Matiere Premiere achetee par saison

QD(D,T) Quantite de Desserts par saison

SMP(MP,T) Stock de Matiere Premiere a la fin de la saison

SD(D,T) Stock de Desserts a la fin de la saison

NOUVR(A,T) Nombre d Ouvriers presents dans chaque atelier par saison

EOUVR(A,T) Nombre d Embauches d ouvriers dans chaque atelier par saison

LOUVR(A,T) Nombre de Licenciements d ouvriers dans chaque atelier par saison

CAMP Cout d Achat des Matieres Premieres

CSTOCK Cout de stockage

CMO Cout de la Main d Oeuvre

CLE Cout des Licenciements et des Engagements

CT Cout Total ;

POSITIVE VARIABLE QMP, QD, SMP, SD, NOUVR, EOUVR, LOUVR, CAMP, CSTOCK, CMO, CLE;

EQUATIONS

COUTMP Cout d achat des matieres premieres

COUTSTOCK Cout de stockage

COUTMO Cout de la main d oeuvre

COUTLE Couts de licenciement et d engagement

COUT Cout total

OFFRE(MP,T) Offre de matiere premiere non depassee

INITSTOCKD(D) Initialisation des stocks de desserts

INITSTOCKMP(MP) Initialisation des stocks de matiere premiere

STOCKMP(MP,T) Stocks de matiere premiere

STOCKD(D,T) Stocks de desserts

OUVRINIT(A) Nombre d ouvriers initiaux

OUVRMAX(A,T) Nombre maximum d ouvriers

OUVRNBR(A,T) Nombre d ouvriers par atelier et par saison

CAPASTOCK(T) Capacite de stockage

HEURES(A,T) Heures disponibles non depassees ;

*Objectif

COUTMP .. CAMP =E= SUM(MP,SUM(T $ T1(T),P(MP,T)*QMP(MP,T)));

COUTSTOCK .. CSTOCK =E= SUM(MP,SUM(T $ T1(T) ,SMP(MP,T)*CSMP(MP)))+SUM(D,SUM(T $ T1(T),SD(D,T)*CSD(D)));

COUTMO .. CMO =E= SUM((A,T),CH(A)*NOUVR(A,T)*J(T)*8);

COUTLE .. CLE =E= SUM(A,SUM(T $ T1(T),CENG(A)*EOUVR(A,T)+CLIC(A)*LOUVR(A,T)));

COUT .. CT =E= CAMP+CSTOCK+CMO+CLE;

*Contraintes

OFFRE(MP,T) $(ORD(T) GT 1) .. QMP(MP,T) =L= OFF(MP,T);

STOCKMP(MP,T) $(ORD(T) GT 1) .. SMP(MP,T) =E= SMP(MP,T-1)+QMP(MP,T)-SUM((D),QI(D,MP)*QD(D,T));

STOCKD(D,T) $(ORD(T) GT 1) .. SD(D,T) =E= SD(D,T-1)+QD(D,T)-DEM(D,T);

OUVRMAX(A,T) $(ORD(T) GT 1) .. NOUVR(A,T) =L= PERSMAX(A);

OUVRNBR(A,T) $(ORD(T) GT 1) .. NOUVR(A,T) =E= NOUVR(A,T-1)+EOUVR(A,T)-LOUVR(A,T);

CAPASTOCK(T) $(ORD(T) GT 1) .. SUM((D),SD(D,T)*VD(D))+SUM((MP),SMP(MP,T)*VMP(MP))=L= CAPA;

HEURES(A,T) $(ORD(T) GT 1) .. SUM((D),(TF(D,A)/60)*QD(D,T)) =E= J(T)*NOUVR(A,T)*8;

*Contraintes - Initialisations

INITSTOCKD(D) .. SD(D,"HIVER0") =E= SID(D);

INITSTOCKMP(MP) .. SMP(MP,"HIVER0") =E= SIMP(MP);

OUVRINIT(A) .. NOUVR(A,"HIVER0") =E= PERSINIT(A);

MODEL TFE /ALL/;

SOLVE TFE USING LP MINIMIZING CT;



Annexe B

Listing de résultats GAMS

(....)

IBM ILOG CPLEX 29.1.0 rbb4180b Released Nov 15, 2019 WEI x86 64bit/MS Window

Cplex 12.9.0.0

Space for names approximately 0.00 Mb

LP status(1): optimal

Cplex Time: 0.00sec (det. 0.16 ticks)

Optimal solution found.

Objective : 428099.306868

LOWER LEVEL UPPER MARGINAL

---- EQU COUTMP . . . 1.0000

---- EQU COUTSTOCK . . . 1.0000

---- EQU COUTMO . . . 1.0000

---- EQU COUTLE . . . 1.0000

---- EQU COUT . . . 1.0000

COUTMP Cout d achat des matieres premieres

COUTSTOCK Cout de stockage

COUTMO Cout de la main d oeuvre

COUTLE Couts de licenciement et d engagement

COUT Cout total

---- EQU OFFRE Offre de matiere premiere non depassee

LOWER LEVEL UPPER MARGINAL

CHOCOLAT .PRINTEMPS -INF 166.5854 +INF .

CHOCOLAT .ETE -INF 300.9146 +INF .

CHOCOLAT .AUTOMNE -INF 350.0000 +INF .

CHOCOLAT .HIVER -INF 300.0000 +INF .

FRUITS_FRAIS.PRINTEMPS -INF 183.7979 1000.0000 .

FRUITS_FRAIS.ETE -INF 771.2021 2000.0000 .

FRUITS_FRAIS.AUTOMNE -INF 500.0000 500.0000 -19.8369

FRUITS_FRAIS.HIVER -INF 60.0000 60.0000 -18.1869

---- EQU INITSTOCKD Initialisation des stocks de desserts

LOWER LEVEL UPPER MARGINAL

DELICES_DU_CACAO 250.0000 250.0000 250.0000 -5.9131

DELICES_DU_VERGER 350.0000 350.0000 350.0000 -7.0800

---- EQU INITSTOCKMP Initialisation des stocks de matiere premiere

LOWER LEVEL UPPER MARGINAL

CHOCOLAT 120.0000 120.0000 120.0000 -6.5000

FRUITS_FRAIS 50.0000 50.0000 50.0000 -1.5000

---- EQU STOCKMP Stocks de matiere premiere

LOWER LEVEL UPPER MARGINAL

CHOCOLAT .PRINTEMPS . . . -6.5000

CHOCOLAT .ETE . . . -6.5000

CHOCOLAT .AUTOMNE . . . -6.5000

CHOCOLAT .HIVER . . . -6.5000

FRUITS_FRAIS.PRINTEMPS . . . -1.5000

FRUITS_FRAIS.ETE . . . -1.0000

FRUITS_FRAIS.AUTOMNE . . . -21.8369

FRUITS_FRAIS.HIVER . . . -22.1869

---- EQU STOCKD Stocks de desserts

LOWER LEVEL UPPER MARGINAL

DELICES_DU_CACAO .PRINTEMPS -4500.0000 -4500.0000 -4500.0000 -5.9131

DELICES_DU_CACAO .ETE -7500.0000 -7500.0000 -7500.0000 -6.1631

DELICES_DU_CACAO .AUTOMNE -7000.0000 -7000.0000 -7000.0000 -8.7125

DELICES_DU_CACAO .HIVER -6000.0000 -6000.0000 -6000.0000 -8.8472

DELICES_DU_VERGER.PRINTEMPS -2500.0000 -2500.0000 -2500.0000 -7.0800

DELICES_DU_VERGER.ETE -1500.0000 -1500.0000 -1500.0000 -7.3300

DELICES_DU_VERGER.AUTOMNE -5000.0000 -5000.0000 -5000.0000 -12.4969

DELICES_DU_VERGER.HIVER -7000.0000 -7000.0000 -7000.0000 -12.7469

---- EQU OUVRINIT Nombre d ouvriers initiaux

LOWER LEVEL UPPER MARGINAL

CONFECTION 30.0000 30.0000 30.0000 2500.0000

CONDITIONNEMENT 12.0000 12.0000 12.0000 1500.0000

---- EQU OUVRMAX Nombre maximum d ouvriers

LOWER LEVEL UPPER MARGINAL

CONFECTION .PRINTEMPS -INF 8.8118 52.0000 .

CONFECTION .ETE -INF 8.8118 52.0000 .

CONFECTION .AUTOMNE -INF 13.7108 52.0000 .

CONFECTION .HIVER -INF 13.7108 52.0000 .

CONDITIONNEMENT.PRINTEMPS -INF 3.2079 22.0000 .

CONDITIONNEMENT.ETE -INF 3.2079 22.0000 .

CONDITIONNEMENT.AUTOMNE -INF 5.1451 22.0000 .

CONDITIONNEMENT.HIVER -INF 5.2808 22.0000 .

---- EQU OUVRNBR Nombre d ouvriers par atelier et par saison

LOWER LEVEL UPPER MARGINAL

CONFECTION .PRINTEMPS . . . 2500.0000

CONFECTION .ETE . . . 682.0488

CONFECTION .AUTOMNE . . . -1000.0000

CONFECTION .HIVER . . . -494.6139

CONDITIONNEMENT.PRINTEMPS . . . 1500.0000

CONDITIONNEMENT.ETE . . . 487.5122

CONDITIONNEMENT.AUTOMNE . . . -500.0000

CONDITIONNEMENT.HIVER . . . -500.0000

---- EQU CAPASTOCK Capacite de stockage

LOWER LEVEL UPPER MARGINAL

PRINTEMPS -INF 853.6411 2000.0000 .

ETE -INF 2000.0000 2000.0000 -8.1948

AUTOMNE -INF 1901.2821 2000.0000 .

HIVER -INF . 2000.0000 .

---- EQU HEURES Heures disponibles non depassees

LOWER LEVEL UPPER MARGINAL

CONFECTION .PRINTEMPS . . . -8.7126

CONFECTION .ETE . . . -9.1626

CONFECTION .AUTOMNE . . . -13.6083

CONFECTION .HIVER . . . -13.5304

CONDITIONNEMENT.PRINTEMPS . . . -7.3907

CONDITIONNEMENT.ETE . . . -7.5407

CONDITIONNEMENT.AUTOMNE . . . -9.5000

CONDITIONNEMENT.HIVER . . . -10.5417

---- VAR QMP Quantite de Matiere Premiere achetee par saison

LOWER LEVEL UPPER MARGINAL

CHOCOLAT .PRINTEMPS . 166.5854 +INF .

CHOCOLAT .ETE . 300.9146 +INF .

CHOCOLAT .AUTOMNE . 350.0000 +INF .

CHOCOLAT .HIVER . 300.0000 +INF .

FRUITS_FRAIS.PRINTEMPS . 183.7979 +INF .

FRUITS_FRAIS.ETE . 771.2021 +INF .

FRUITS_FRAIS.AUTOMNE . 500.0000 +INF .

FRUITS_FRAIS.HIVER . 60.0000 +INF .

---- VAR QD Quantite de Desserts par saison

LOWER LEVEL UPPER MARGINAL

DELICES_DU_CACAO .PRINTEMPS . 5731.7073 +INF .

DELICES_DU_CACAO .ETE . 6018.2927 +INF .

DELICES_DU_CACAO .AUTOMNE . 7000.0000 +INF .

DELICES_DU_CACAO .HIVER . 6000.0000 +INF .

DELICES_DU_VERGER.PRINTEMPS . 2337.9791 +INF .

DELICES_DU_VERGER.ETE . 2454.8780 +INF .

DELICES_DU_VERGER.AUTOMNE . 4717.9487 +INF .

DELICES_DU_VERGER.HIVER . 6139.1941 +INF .

---- VAR SMP Stock de Matiere Premiere a la fin de la saison

LOWER LEVEL UPPER MARGINAL

CHOCOLAT .HIVER0 . 120.0000 +INF .

CHOCOLAT .PRINTEMPS . . +INF 0.1000

CHOCOLAT .ETE . . +INF 0.1000

CHOCOLAT .AUTOMNE . . +INF 0.1000

CHOCOLAT .HIVER . . +INF 6.6000

FRUITS_FRAIS.HIVER0 . 50.0000 +INF .

FRUITS_FRAIS.PRINTEMPS . . +INF 0.8500

FRUITS_FRAIS.ETE . 525.7143 +INF .

FRUITS_FRAIS.AUTOMNE . 553.9194 +INF .

FRUITS_FRAIS.HIVER . . +INF 22.5369

---- VAR SD Stock de Desserts a la fin de la saison

LOWER LEVEL UPPER MARGINAL

DELICES_DU_CACAO .HIVER0 . 250.0000 +INF .

DELICES_DU_CACAO .PRINTEMPS . 1481.7073 +INF .

DELICES_DU_CACAO .ETE . . +INF 1.7980

DELICES_DU_CACAO .AUTOMNE . . +INF 0.1153

DELICES_DU_CACAO .HIVER . . +INF 9.0972

DELICES_DU_VERGER.HIVER0 . 350.0000 +INF .

DELICES_DU_VERGER.PRINTEMPS . 187.9791 +INF .

DELICES_DU_VERGER.ETE . 1142.8571 +INF .

DELICES_DU_VERGER.AUTOMNE . 860.8059 +INF .

DELICES_DU_VERGER.HIVER . . +INF 12.9969

---- VAR NOUVR Nombre d Ouvriers presents dans chaque atelier par saison

LOWER LEVEL UPPER MARGINAL

CONFECTION .HIVER0 . 30.0000 +INF .

CONFECTION .PRINTEMPS . 8.8118 +INF .

CONFECTION .ETE . 8.8118 +INF .

CONFECTION .AUTOMNE . 13.7108 +INF .

CONFECTION .HIVER . 13.7108 +INF .

CONDITIONNEMENT.HIVER0 . 12.0000 +INF .

CONDITIONNEMENT.PRINTEMPS . 3.2079 +INF .

CONDITIONNEMENT.ETE . 3.2079 +INF .

CONDITIONNEMENT.AUTOMNE . 5.1451 +INF .

CONDITIONNEMENT.HIVER . 5.2808 +INF .

---- VAR EOUVR Nombre d Embauches d ouvriers dans chaque atelier par saison

LOWER LEVEL UPPER MARGINAL

CONFECTION .PRINTEMPS . . +INF 3500.0000

CONFECTION .ETE . . +INF 1682.0488

CONFECTION .AUTOMNE . 4.8990 +INF .

CONFECTION .HIVER . . +INF 505.3861

CONDITIONNEMENT.PRINTEMPS . . +INF 2000.0000

CONDITIONNEMENT.ETE . . +INF 987.5122

CONDITIONNEMENT.AUTOMNE . 1.9372 +INF .

CONDITIONNEMENT.HIVER . 0.1358 +INF .

---- VAR LOUVR Nombre de Licenciements d ouvriers dans chaque atelier par saison

LOWER LEVEL UPPER MARGINAL

CONFECTION .PRINTEMPS . 21.1882 +INF .

CONFECTION .ETE . . +INF 1817.9512

CONFECTION .AUTOMNE . . +INF 3500.0000

CONFECTION .HIVER . . +INF 2994.6139

CONDITIONNEMENT.PRINTEMPS . 8.7921 +INF .

CONDITIONNEMENT.ETE . . +INF 1012.4878

CONDITIONNEMENT.AUTOMNE . . +INF 2000.0000

CONDITIONNEMENT.HIVER . . +INF 2000.0000

LOWER LEVEL UPPER MARGINAL

---- VAR CAMP . 9550.6490 +INF .

---- VAR CSTOCK . 1296.2091 +INF .

---- VAR CMO . 345158.3333 +INF .

---- VAR CLE . 72094.1154 +INF .

---- VAR CT -INF 428099.3069 +INF .

CAMP Cout d Achat des Matieres Premieres

CSTOCK Cout de stockage

CMO Cout de la Main d Oeuvre

CLE Cout des Licenciements et des Engagements

CT Cout Total


